


Beatriz Rocha A84003



Filipe Guimarães A85308



Gonçalo Ferreira A84073

Relatório do Trabalho Prático 1 de Métodos Determinísticos de Investigação Operacional Grupo 32

28 de Outubro de 2020



Conteúdo

1	Introdução	3		
	1.1 Contextualização	3		
	1.2 Objetivos	3		
2	Formulação do Modelo	4		
	2.1 Mapa da Cidade	4		
	2.2 Variáveis de decisão	4		
	2.3 Dados	5		
	2.4 Restrições	5		
	2.5 Função objetivo	5		
	2.6 Dimensão dos elementos do modelo	5		
3	Ficheiro de input	6		
4	Ficheiro de output	8		
5	Interpretação dos resultados	9		
6	Validação do modelo	12		
7	Dificuldades surgidas durante a realização do trabalho			
8	Conclusão	14		

Lista de Figuras

Sentido das ruas BCDE	4
Primeira volta	ç
Segunda volta	(
Terceira volta	(
Quarta volta	1
Percurso completo	1
	Primeira volta

Introdução

1.1 Contextualização

Um veículo de recolha de lixo tem de percorrer ruas de sentido único para recolher os sacos existentes ao longo das mesmas. Assim sendo, pretende-se determinar o circuito ou conjunto de circuitos em que todos os arcos de um grafo são percorridos, pelo menos, uma vez, minimizando a distância total percorrida.

1.2 Objetivos

O principal objetivo deste trabalho prático consiste em resolver o problema descrito acima, aplicando os conhecimentos lecionados nas aulas, nomeadamente modelos de programação linear.

Formulação do Modelo

2.1 Mapa da Cidade

Tendo em conta que 85308 é o número do aluno do grupo com maior número de inscrição, o sentido da rua B é a subir, o sentido da rua C é para a esquerda, o sentido da rua D é a descer e o sentido da rua E é para a direita, tal como podemos ver na figura 2.1.




Figura 2.1: Sentido das ruas BCDE

2.2 Variáveis de decisão

Tendo em consideração que existem 14 nodos no mapa e nem todos estão relacionados diretamente uns com os outros, as nossas variáveis de decisão serão os vários arcos possíveis do grafo, representando as ruas da cidade (à exceção dos arcos x7.1 e x1.2 que formam uma única rua), nomeadamente x1.2, x2.3,

 $x2_{-9}$, $x3_{-4}$, $x4_{-5}$, $x4_{-13}$, $x5_{-6}$, $x6_{-7}$, $x7_{-1}$, $x8_{-7}$, $x8_{-10}$, $x9_{-8}$, $x10_{-9}$, $x10_{-11}$, $x11_{-3}$, $x11_{-6}$, $x12_{-11}$, $x12_{-14}$, $x13_{-12}$, $x14_{-5}$ e $x14_{-13}$.

2.3 Dados

Os dados deste problema são:

- Comprimento dos arcos;
- Sentido dos arcos;
- Posição dos nodos;
- Ponto de partida do veículo de recolha de lixo;
- Ponto de chegada do veículo de recolha de lixo.

2.4 Restrições

Uma vez que o veículo de recolha de lixo tem de percorrer cada um dos arcos, pelo menos, uma vez e dado que o número de vezes que este entra num nodo tem de ser igual ao número de vezes que sai, teremos de satisfazer estas restrições.

2.5 Função objetivo

O principal objetivo deste problema de programação linear é minimizar a distância que o veículo de recolha de lixo tem de percorrer, de forma a atravessar cada uma das ruas de sentido único, pelo menos, uma vez. Assim sendo, teremos de minimizar a nossa função objetivo que se traduzirá na soma das distâncias associadas a cada arco.

2.6 Dimensão dos elementos do modelo

Comecemos pelas variáveis de decisão. Cada variável de decisão xa_b representará o número de passagens no arco a_b por trajeto e, portanto, a sua dimensão será [passagem_arco/percurso].

Passemos aos parâmetros. Cada parâmetro representará a distância percorrida por cada passagem no arco correspondente e, assim, a sua dimensão será [distância/passagem_arco].

Daqui, tiramos que a dimensão da função objetivo é [distância/percurso], o que é consistente, uma vez que o seu resultado representará a distância percorrida por cada percurso efetuado.

Por último, falemos das restrições. A dimensão do primeiro grupo de restrições será [passagem_arco/percurso], pois queremos que o veículo de recolha de lixo passe pelo menos uma vez em cada arco por cada percurso. Já a dimensão do segundo grupo de restrições será também [passagem_arco/percurso], uma vez que desejamos que o número de passagens nos arcos à esquerda da igualdade por cada percurso seja o mesmo que o número de passagens nos arcos à direita da igualdade por cada percurso.

Ficheiro de input

A função objetivo apresenta-se de seguida:

```
min: 3 x1_2 + 3 x2_3 + 3 x2_9 + 4 x3_4 + 12 x4_5
+ 2 x4_13 + 4 x5_6 + 4 x6_7 + 10 x7_1
+ 2 x8_7 + 4 x8_10 + 6 x9_8 + 2 x10_9
+ 2 x10_11 + 4 x11_3 + 4 x11_6 + 3 x12_11
+ 2 x12_14 + 3 x13_12 + 3 x14_5 + 5 x14_13;
```

As restrições que forçam a passagem do veículo de recolha de lixo, pelo menos, uma vez em cada arco são as seguintes:

```
x1_2 >= 1;
x2_3 >= 1;
x2_9 >= 1;
x3_4 >= 1;
x4_5 >= 1;
x4_13 >= 1;
x5_6 >= 1;
x6_7 >= 1;
x7_1 >= 1;
x8_7 >= 1;
x8_10 >= 1;
x9_8 >= 1;
x10_9 >= 1;
x10_11 >= 1;
x11_3 >= 1;
x11_6 >= 1;
x12_11 >= 1;
x12_14 >= 1;
x13_12 >= 1;
x14_5 >= 1;
x14_13 >= 1;
```

Uma vez que o número de vezes que o veículo de recolha de lixo entra num nodo tem de ser igual ao número de vezes que sai, é necessário colocar as seguintes restrições:

```
x1_2 = x7_1;

x2_9 + x2_3 = x1_2;

x3_4 = x2_3 + x11_3;

x4_13 + x4_5 = x3_4;

x5_6 = x4_5 + x14_5;

x6_7 = x5_6 + x11_6;

x7_1 = x6_7 + x8_7;

x8_10 + x8_7 = x9_8;

x9_8 = x2_9 + x10_9;

x10_9 + x10_11 = x8_10;

x11_3 + x11_6 = x10_11 + x12_11;

x12_11 + x12_14 = x13_12;

x13_12 = x4_13 + x14_13;

x14_5 + x14_13 = x12_14;
```

Ficheiro de output

A solução ótima deste problema é 176.

Os valores das variáveis são:

x1_2	4
x2_3	2
x2_9	2
x3_4	3
x4_5	1
x4_13	2
x5_6	2
x6_7	3
x7_1	4
x8_7	1
x8_10	2
x9_8	3
x10_9	1
x10_11	1
x11_3	1
x11_6	1
x12_11	1
x12_14	2
x13_12	3
x14_5	1
x14_13	1

Interpretação dos resultados

A solução ótima indica-nos o número de vezes que o veículo de recolha de lixo percorre cada arco. Deste modo, teremos o percurso indicado nas figuras abaixo.




Figura 5.1: Primeira volta




Figura 5.2: Segunda volta




Figura 5.3: Terceira volta




Figura 5.4: Quarta volta




Figura 5.5: Percurso completo

Substituindo os valores da solução ótima na função objetivo, tiramos que o custo deste percurso é 176, isto é, 176 é a distância que minimiza a função objetivo, satisfazendo todas as restrições.

Validação do modelo

Observando o ficheiro de output e desenhando o percurso no mapa, chegamos à conclusão que a solução é válida, uma vez que cada arco é percorrido, pelo menos, uma vez e o número de vezes que o veículo de recolha de lixo entra em cada nodo é igual ao número de vezes que sai, satisfazendo, assim, os grupos de restrições.

Dificuldades surgidas durante a realização do trabalho

A maior dificuldade surgida durante a realização deste trabalho consistiu na escolha das variáveis de decisão. Inicialmente, pensamos em escolher as arestas da figura 2.1 como variáveis de decisão. Contudo, acreditamos que a escolha dos arcos como variáveis de decisão é mais vantajosa, na medida em que é menos trabalhoso e conseguimos facilmente observar quais os arcos que estão relacionados.

Conclusão

Em resumo, com este projeto, tivemos oportunidade de consolidar e pôr em prática os conhecimentos adquiridos nesta unidade curricular, numa situação do mundo real. Acreditamos ter sido bastante enriquecedor na medida em que nos poderão surgir vários problemas deste cariz no nosso dia-a-dia.